# Maintainer: Antonio Rojas <arojas@archlinux.org>
# Contributor: Jan de Groot <jgc@archlinux.org>

pkgname=libspectre
pkgver=0.2.12
pkgrel=1
pkgdesc='Small library for rendering Postscript documents'
arch=(x86_64)
license=(GPL2)
url='https://www.freedesktop.org/wiki/Software/libspectre'
depends=(ghostscript)
source=(https://libspectre.freedesktop.org/releases/$pkgname-$pkgver.tar.gz{,.sig})
sha256sums=('55a7517cd3572bd2565df0cf450944a04d5273b279ebb369a895391957f0f960'
            'SKIP')
validpgpkeys=(CA262C6C83DE4D2FB28A332A3A6A4DB839EAA6D7) # Albert Astals Cid <aacid@kde.org>

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr --disable-static
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}
